Math 277
Ordinary Differential Equations

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 1
Introduction

Initializations

```
> restart;
  with(plots):
```

1.1 Some Basic Techniques

Examples

Example 1.1.1

Show that $\phi(x) = 3x^2 - 4x^{-1}$ is an explicit solution to the linear equation

$$\frac{d^2y}{dx^2} - \frac{2}{x^2}y = 0$$

Solution

Code the function ϕ and the left hand side of the differential equation.

```
> phi:=x->3*x^2-4*x^(-1);
```

$$\phi := x \to 3x^2 - \frac{4}{x} \qquad \textbf{(2.1.1.1)}$$

```
> L:=diff(y(x), x$2)-2/x^2*y(x);
```

$$L := \frac{d^2}{dx^2}y(x) - \frac{2y(x)}{x^2} \qquad \textbf{(2.1.1.2)}$$

We show that replacing y by ϕ in L, and simplification of the result, yields the value zero

```
> e1:=eval(L, y=phi);
  e2:=simplify(e1);
```

$$e1 := 6 - \frac{8}{x^3} - \frac{2\left(3x^2 - \frac{4}{x}\right)}{x^2}$$

$$e2 := 0 \qquad \textbf{(2.1.1.3)}$$

Hence the function ϕ is an explicit solution of the given differential equation.

Example 1.1.2

Verify that for every constant C the relation $4x^2 - y^2 = C$ is an implicit solution to

$$y\,\frac{dy}{dx} - 4x = 0$$

Graph the solution curves for $C = 0, \pm 1, \pm 4$.

Solution

Code the implicitly defined curve, and differentiate the result with respect to the variable x.

```
> sol:=4*x^2-y^2=C;
  sol_x:=subs(y=y(x), sol);
```

$$sol := 4x^2 - y^2 = C$$

$$sol\_x := 4x^2 - y(x)^2 = C \qquad \textbf{(2.1.2.1)}$$

```
> e1:=diff(sol_x, x);
```

$$e1 := 8x - 2y(x)\left(\frac{d}{dx}y(x)\right) = 0 \qquad \textbf{(2.1.2.2)}$$

Division by -2 yields the given differential equation. To produce the desired graph, we first create a list containing the solution curves.

```
> sc:=[seq(sol, C=[0,-1,1,-4,4])];
```

$$sc := [4x^2 - y^2 = 0, 4x^2 - y^2 = -1, 4x^2 - y^2 = 1, 4x^2 - y^2 = -4, 4x^2 - y^2 = 4] \qquad \textbf{(2.1.2.3)}$$

We plot these implicit functions using Maple's **implicitplot** routine.

```
> implicitplot(sc, x=-2..2, y=-3..3, color=[red, blue,
  green, magenta, gold], numpoints=8000, scaling=
  constrained);
```



Compare this graphic with Figure 1.4 on Page 9 of the textbook.